

Example 6c: Modeling Change over Time Using Truly Exponential Trends (complete data, syntax, and output available for SAS electronically)

These data for these example models come from Hoffman (2015) chapter 6. We will be examining change in response time (RT) in milliseconds over six practice sessions (balanced time) to a measure of processing speed in a sample of 101 older adults. This Example 6c builds on polynomial time slopes (previously examined in Example 6a) and piecewise slopes for initial change and later change (Example 6b) to specify “truly” nonlinear exponential models. Stay tuned for the use of log-transformed time to approximate a (truly nonlinear) exponential model (Example 6d).

SAS Syntax for Data Import and Manipulation:

```
* Define global variable for file location to be replaced in code below;
%LET filesave=C:\Dropbox\21_PSQF7375_Longitudinal\PSQF7375_Longitudinal_Example6;
* Location for SAS files for these models (uses macro variable filesave);
LIBNAME filesave "&filesave.';

DATA work.Example6; SET filesave.SAS_Chapter6;
* Center time predictor for polynomial time models;
time1=session-1; LABEL time1="time1: Session (0=1)";
* Create session dummy codes for testing means model absolute fit;
IF session=1 THEN s1=1; ELSE s1=0;
IF session=2 THEN s2=1; ELSE s2=0;
IF session=3 THEN s3=1; ELSE s3=0; RUN;
```

Truly nonlinear models (like the exponential) will require learning a new procedure—SAS NLMIXED. REML estimation is not available in NLMIXED, so these models will be estimated using ML (with additional options given below) instead. To get started with NLMIXED, we begin by using it to specify two familiar models (previously estimated using MIXED): an empty means, random intercept model (1b), and a random quadratic time model (3b).

Model 1b. Empty Means, Random Intercept Model via MIXED and NLMIXED

$$\text{Level 1: } y_{ti} = \beta_{0i} + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

```
TITLE1 "SAS Model 1b: Empty Means, Random Intercept Model using ML";
PROC MIXED DATA=work.Example6 NOCLPRINT COVTEST METHOD=ML;
  CLASS PersonID session;
  MODEL rt = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
RUN; TITLE1;
```

In the PROC NLMIXED line below, adaptive Gauss-Hermite Quadrature (METHOD=GAUSS) is used to integrate over random effects (necessary for non-normal outcomes, but not really relevant here given that our random effects and residuals here are each assumed to be normally distributed.). Newton–Raphson optimization (TECH=NEWRAP) is a specific way of finding the top of the likelihood mountain. Finally, we also set stricter gradient convergence criteria (GCONV=1e-12) to help ensure that each parameter is really at the top of its dimension of the likelihood mountain.

```
TITLE1 "SAS Model 1b: Empty Means, Random Intercept Model via NLMIXED";
PROC NLMIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances and covariances;
PARMS fint=1770
      VarU0=198820 VarE=44900;
* Setting up level-2 equations;
b0i = fint + U0i;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i);
* Telling it which DV, defining level-1 residual;
* RT is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY, VarE);
```

```
* Random effects are normally distributed with means=0 and estimated variances;
  RANDOM U0i ~ normal([0],[VarU0]) SUBJECT=PersonID;
* Asking for ICC and its SE;
  ESTIMATE "ICC" VarU0 / (VarU0 + VarE);
RUN; TITLE1;
```

One advantage that NLMIXED has is that it can provide linear combinations of variance model parameters (as well as mean model parameters). Here, I use ESTIMATE to request the ICC and its SE.

MIXED OUTPUT:

Covariance Parameter Estimates					
		Subject	Estimate	Standard Error	Z
Cov Parm	UN(1,1)	PersonID	198820	29035	6.85 <.0001
Session		PersonID	44900	2825.63	15.89 <.0001
Fit Statistics					
-2 Log Likelihood			8546.3		
AIC (smaller is better)			8552.3		
AICC (smaller is better)			8552.4		
BIC (smaller is better)			8560.2		

Solution for Fixed Effects

	Effect	Estimate	Error	DF	t Value	Pr > t
Intercept		1770.70	45.1952	101	39.18	<.0001

NLMIXED OUTPUT:

Fit Statistics	
-2 Log Likelihood	8546.3
AIC (smaller is better)	8552.3
AICC (smaller is better)	8552.4
BIC (smaller is better)	8560.2

The “gradient” column provides the partial first derivatives of the LL function with respect to each parameter. You want these to be as close to 0 as possible, otherwise they are not trustworthy (i.e., it’s not really at the top of its dimension of the LL mountain).

Parameter Estimates										
	Parameter	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fint	1770.69	45.1952	100	39.18	<.0001	0.05	1681.02	1860.36	-5.4E-6	
VarU0	198820	29035	100	6.85	<.0001	0.05	141216	256424	2.82E-10	
VarE	44900	2825.64	100	15.89	<.0001	0.05	39294	50506	3.324E-9	

Additional Estimates									
	Label	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper
ICC		0.8158	0.02404	100	33.94	<.0001	0.05	0.7681	0.8635

Model 3b. Random Quadratic Time Model via MIXED and NLMIXED

```
TITLE1 "SAS Model 3b: Random Quadratic Time Model using ML";
PROC MIXED DATA=work.Example6 NOCLPRINT COVTEST METHOD=ML;
  CLASS PersonID session;
  MODEL rt = time1 time1*time1 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time1 time1*time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
  REPEATED session / R TYPE=VC SUBJECT=PersonID;
RUN; TITLE1;

TITLE1 "SAS Model 3b: Random Quadratic Time Model via NLMIXED";
PROC NLMIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second and third lines are variances and covariances;
  PARMS fint=1946 flin=-121 fquad=14
        VarU0=273306 CovU10=-35626 VarU1=25438
        CovU20=3845 CovU21=-3838 VarU2=622 VarE=20298;
```

```

* Setting up level-2 equations;
b0i = fint + U0i;
b1i = flin + U1i;
b2i = fquad + U2i;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*time1) + (b2i*time1*time1);
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY, VarE);
* Random effects are normally distributed with means=0 and estimated variances/covariances;
RANDOM U0i U1i U2i ~ normal([0,0,0],[VarU0,CovU10,VarU1,CovU20,CovU21,VarU2]) SUBJECT=PersonID;
RUN; TITLE1;

```

MIXED OUTPUT:

Estimated G Matrix					
Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	273306	-35262	3845.38
2	time1	101	-35262	25438	-3837.76
3	time1*time1	101	3845.38	-3837.76	622.81

Covariance Parameter Estimates					
Cov	Parm	Subject	Estimate	Standard	Z
UN(1,1)	PersonID		273306	40828	6.69 <.0001
UN(2,1)	PersonID		-35262	11765	-3.00 0.0027
UN(2,2)	PersonID		25438	5781.19	4.40 <.0001
UN(3,1)	PersonID		3845.38	1920.35	2.00 0.0452
UN(3,2)	PersonID		-3837.76	968.79	-3.96 <.0001
UN(3,3)	PersonID		622.81	169.99	3.66 0.0001
Session	PersonID		20298	1649.11	12.31 <.0001

Fit Statistics					
-2 Log Likelihood			8321.8		
AIC (smaller is better)			8341.8		
AICC (smaller is better)			8342.1		
BIC (smaller is better)			8367.9		

Solution for Fixed Effects					
Standard					
Effect	Estimate	Error	DF	t Value	Pr > t
Intercept	1945.85	53.5825	101	36.32	<.0001
time1	-120.90	19.9481	101	-6.06	<.0001
time1*time1	13.8656	3.3985	101	4.08	<.0001

NLMIXED OUTPUT:

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics									
-2 Log Likelihood									8321.8
AIC (smaller is better)									8341.8
AICC (smaller is better)									8342.1
BIC (smaller is better)									8367.9
Parameter Estimates									
Standard									
Parameter	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fint	1945.85	53.5832	98	36.31	<.0001	0.05	1839.52	2052.18	1.81E-10
flin	-120.90	19.9571	98	-6.06	<.0001	0.05	-160.50	-81.2957	8.76E-11
fquad	13.8656	3.3987	98	4.08	<.0001	0.05	7.1210	20.6102	2.24E-10

VarU0	273313	40831	98	6.69	<.0001	0.05	192286	354341	-6.15E-7
CovU10	-35568	11799	98	-3.01	0.0033	0.05	-58982	-12154	-4.28E-6
VarU1	25474	5791.26	98	4.40	<.0001	0.05	13981	36966	-3.04E-6
CovU20	3898.39	1923.78	98	2.03	0.0454	0.05	80.7058	7716.07	1.039E-8
CovU21	-3841.28	969.79	98	-3.96	0.0001	0.05	-5765.79	-1916.77	-3E-7
VarU2	622.95	170.04	98	3.66	0.0004	0.05	285.52	960.38	-2.74E-6
VarE	20299	1649.21	98	12.31	<.0001	0.05	17026	23572	-8.05E-9

Because variances can be hard to estimate, the negative exponential models that follow instead estimate standard deviations (SD) directly, and then calculate variances by squaring those SDs within ESTIMATE statements.

Model 6a. Negative Exponential Model (Fixed Asymptote, Fixed Amount, Fixed Rate)

```
TITLE1 "SAS Ch 6: 6a Fixed Asymptote Fixed Amount Fixed Rate Model";
PROC NLIN MIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasymp=1600 famount=300 frate=-1
      SDE=600;
* Setting up level-2 equations;
b0i = fasymp;
b1i = famount;
b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*time));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY, sdE*sde);
* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"          fasymp;
ESTIMATE "Fixed Amount"             famount;
ESTIMATE "Fixed Rate"              frate;
ESTIMATE "Level-1 Residual E Variance" sde*sde;
* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"         fasymp+famount;
ESTIMATE "Session=1 Time=0 Predicted Mean" fasymp+(famount*EXP(frate*0));
ESTIMATE "Session=2 Time=1 Predicted Mean" fasymp+(famount*EXP(frate*1));
ESTIMATE "Session=3 Time=2 Predicted Mean" fasymp+(famount*EXP(frate*2));
ESTIMATE "Session=4 Time=3 Predicted Mean" fasymp+(famount*EXP(frate*3));
ESTIMATE "Session=5 Time=4 Predicted Mean" fasymp+(famount*EXP(frate*4));
ESTIMATE "Session=6 Time=5 Predicted Mean" fasymp+(famount*EXP(frate*5));
RUN; TITLE1;
```

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics																	
-2 Log Likelihood		9213.2															
AIC (smaller is better)		9221.2															
AICC (smaller is better)		9221.2															
BIC (smaller is better)		9238.8															
Parameter Estimates																	
Standard																	
Parameter	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient								
fasymp	1675.25	54.8326	606	30.55	<.0001	0.05	1567.56	1782.93	1.069E-9								
famount	284.71	64.5965	606	4.41	<.0001	0.05	157.85	411.57	3.15E-10								
frate	-0.6698	0.4247	606	-1.58	0.1153	0.05	-1.5039	0.1643	1.704E-7								
SDE	484.28	13.9107	606	34.81	<.0001	0.05	456.97	511.60	-336E-13								
Additional Estimates																	
Standard																	
Label	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper									
Fixed Asymptote	1675.25	54.8326	606	30.55	<.0001	0.05	1567.56	1782.93									
Fixed Amount	284.71	64.5965	606	4.41	<.0001	0.05	157.85	411.57									
Fixed Rate	-0.6698	0.4247	606	-1.58	0.1153	0.05	-1.5039	0.1643									
Residual E Variance	234532	13474	606	17.41	<.0001	0.05	208071	260992									

Fixed Intercept	1959.96	47.8094	606	41.00	<.0001	0.05	1866.07	2053.85
Session=1 Time=0 Predicted Mean	1959.96	47.8094	606	41.00	<.0001	0.05	1866.07	2053.85
Session=2 Time=1 Predicted Mean	1820.97	36.2937	606	50.17	<.0001	0.05	1749.69	1892.24
Session=3 Time=2 Predicted Mean	1749.83	30.7816	606	56.85	<.0001	0.05	1689.38	1810.28
Session=4 Time=3 Predicted Mean	1713.42	23.4669	606	73.01	<.0001	0.05	1667.33	1759.51
Session=5 Time=4 Predicted Mean	1694.79	27.9049	606	60.73	<.0001	0.05	1639.98	1749.59
Session=6 Time=5 Predicted Mean	1685.25	36.3828	606	46.32	<.0001	0.05	1613.80	1756.70

Model 6b. Negative Exponential Model (Add Random Asymptote, Fixed Amount, Fixed Rate)

```

TITLE1 "SAS Ch 6: 6b Random Asymptote Fixed Amount Fixed Rate Model";
PROC NLINMIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasympt= 1675 famount=284 frate=-.7
      sdE=474 sdU0=10;
* Setting up level-2 equations;
b0i = fasympt + U0i;
b1i = famount;
b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*time1));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY, sdE*sdE);
* Defining random effects: normally distributed with means and variances;
RANDOM U0i ~ normal([0],[sdU0*sdU0]) SUBJECT=PersonID;
* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"          fasympt;
ESTIMATE "Fixed Intercept"          fasympt+famount;
ESTIMATE "Fixed Amount"             famount;
ESTIMATE "Fixed Rate"              frate;
ESTIMATE "Level-1 Residual E Variance" sdE*sdE;
ESTIMATE "Level-2 Random Asymptote U0 Variance" sdU0*sdU0;
* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"          fasympt+famount;
ESTIMATE "Session=1 Time=0 Predicted Mean" fasympt+(famount*EXP(frate*0));
ESTIMATE "Session=2 Time=1 Predicted Mean" fasympt+(famount*EXP(frate*1));
ESTIMATE "Session=3 Time=2 Predicted Mean" fasympt+(famount*EXP(frate*2));
ESTIMATE "Session=4 Time=3 Predicted Mean" fasympt+(famount*EXP(frate*3));
ESTIMATE "Session=5 Time=4 Predicted Mean" fasympt+(famount*EXP(frate*4));
ESTIMATE "Session=6 Time=5 Predicted Mean" fasympt+(famount*EXP(frate*5));
RUN; TITLE1;

```

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	8404.0
AIC (smaller is better)	8414.0
AICC (smaller is better)	8414.1
BIC (smaller is better)	8427.1

The random asymptote variance is significant relative to the previous model, $-2\Delta LL(1) = 809.12, p < .0001$.

Parameter Estimates									
Parameter	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fasympt	1675.25	49.2032	100	34.05	<.0001	0.05	1577.63	1772.87	3.31E-8
famount	284.71	24.5497	100	11.60	<.0001	0.05	236.00	333.41	1.102E-8
frate	-0.6698	0.1614	100	-4.15	<.0001	0.05	-0.9900	-0.3495	2.948E-6
sdE	184.05	5.7913	100	31.78	<.0001	0.05	172.56	195.54	-1.55E-7
sdU0	447.95	32.4064	100	13.82	<.0001	0.05	383.65	512.24	-1.59E-6
Additional Estimates									
Label	Estimate	Error	DF	t Value	Pr > t	Alpha	Lower	Upper	
Fixed Asymptote	1675.25	49.2032	100	34.05	<.0001	0.05	1577.63	1772.87	
Fixed Intercept	1959.96	48.1335	100	40.72	<.0001	0.05	1864.46	2055.45	
Fixed Amount	284.71	24.5497	100	11.60	<.0001	0.05	236.00	333.41	
Fixed Rate	-0.6698	0.1614	100	-4.15	<.0001	0.05	-0.9900	-0.3495	
Residual E Variance	33875	2131.79	100	15.89	<.0001	0.05	29645	38104	

Random Asymptote U0 Variance 200656 29033 100 6.91 <.0001 0.05 143056 258256

Model 6c. Negative Exponential Model (Random Asymptote, Add Random Amount, Fixed Rate)

```

TITLE1 "SAS Ch 6: 6c Random Asymptote Random Amount Fixed Rate Model";
PROC NLIN MIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasypmp=1675 famount=284 frate=-.7
      sdE=184 sdU0=447 sdU01=1 sdU1=10;
* Setting up level-2 equations;
b0i = fasypmp + U0i;
bli = famount + Uli;
b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (bli*EXP(b2i*time1));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "y" and a variance of "VarE";
MODEL rt ~ normal(PredY, sdE*sdE);
* Defining random effects: normally distributed with means and variances;
RANDOM U0i Uli ~ normal([0,0],[sdU0*sdU0,sdU01*sdU01, sdU1*sdU1]) SUBJECT=PersonID;
* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote" fasypmp;
ESTIMATE "Fixed Amount" famount;
ESTIMATE "Fixed Rate" frate;
ESTIMATE "Level-1 Residual E Variance" sdE*sdE;
ESTIMATE "Level-2 Random Asymptote U0 Variance" sdU0*sdU0;
ESTIMATE "Level-2 Asymptote-Amount U01 Covariance" sdU01*sdU01;
ESTIMATE "Level-2 Random Amount U1 Variance" sdU1*sdU1;
ESTIMATE "Level-2 Asymptote-Amount Correlation" (sdU01*sdU01)/(sdU0*sdU1);
* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept" fasypmp+famount;
ESTIMATE "Session=1 Time=0 Predicted Mean" fasypmp+(famount*EXP(frate*0));
ESTIMATE "Session=2 Time=1 Predicted Mean" fasypmp+(famount*EXP(frate*1));
ESTIMATE "Session=3 Time=2 Predicted Mean" fasypmp+(famount*EXP(frate*2));
ESTIMATE "Session=4 Time=3 Predicted Mean" fasypmp+(famount*EXP(frate*3));
ESTIMATE "Session=5 Time=4 Predicted Mean" fasypmp+(famount*EXP(frate*4));
ESTIMATE "Session=6 Time=5 Predicted Mean" fasypmp+(famount*EXP(frate*5));
RUN;

```

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics		
-2 Log Likelihood		8327.3
AIC (smaller is better)		8341.3
AICC (smaller is better)		8341.5
BIC (smaller is better)		8359.6

The random amount variance is significant relative to the previous model, $-2\Delta LL(2) = 76.69, p < .0001$.

Parameter Estimates

Parameter	Estimate	Error	DF	t Value	Pr > t	Alpha	Standard		Gradient
							Lower	Upper	
fasypmp	1683.48	45.4523	99	37.04	<.0001	0.05	1593.30	1773.67	-3.75E-6
famount	279.94	33.5457	99	8.35	<.0001	0.05	213.38	346.51	-1.59E-7
frate	-0.7533	0.1181	99	-6.38	<.0001	0.05	-0.9877	-0.5189	5.535E-6
sdE	151.79	5.3422	99	28.41	<.0001	0.05	141.19	162.39	-2.98E-7
sdU0	436.83	31.8997	99	13.69	<.0001	0.05	373.54	500.13	-6.18E-7
sdU01	81.5157	90.6707	99	0.90	0.3708	0.05	-98.3947	261.43	-3.39E-6
sdU1	277.95	28.4727	99	9.76	<.0001	0.05	221.45	334.44	-1.86E-7

Additional Estimates

Label	Estimate	Error	DF	t Value	Pr > t	Alpha	Standard		Upper
							Lower	Upper	
Fixed Asymptote	1683.48	45.4523	99	37.04	<.0001	0.05	1593.30	1773.67	
Fixed Amount	279.94	33.5457	99	8.35	<.0001	0.05	213.38	346.51	
Fixed Rate	-0.7533	0.1181	99	-6.38	<.0001	0.05	-0.9877	-0.5189	
Residual E Variance	23039	1621.75	99	14.21	<.0001	0.05	19821	26257	
Random Asymptote U0 Variance	190823	27870	99	6.85	<.0001	0.05	135523	246122	
Asymptote-Amount U01 Covariance	6644.80	14782	99	0.45	0.6540	0.05	-22686	35976	
Random Amount U1 Variance	77254	15828	99	4.88	<.0001	0.05	45848	108659	
Asymptote-Amount Correlation	0.05473	0.1224	99	0.45	0.6557	0.05	-0.1881	0.2975	

95% Random Effect Confidence Intervals that describe the *predicted* range of *individual* random effects:

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm (1.96 * \sqrt{\text{Random Variance}})$$

$$\text{Asymptote 95\% CI} = \gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,683.5 \pm (1.96 * \sqrt{190,823}) = 827 \text{ to } 2,540$$

$$\text{Amount 95\% CI} = \gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow 279.9 \pm (1.96 * \sqrt{77,254}) = -265 \text{ to } 825$$

Model 6d. Negative Exponential Model (Random Asymptote and Amount, Add Random Rate)

```

TITLE1 "SAS Ch 6: 6d Random Asymptote Random Amount Random Rate Model";
TITLE2 "No convergence";
PROC NLIN MIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasypmp=1678.25 famount=-282.72 frate=-.7323
      sdE=130.88 sdU0=426.79 sdU01=87.68 sdU1=290.57
      sdU02=5.1343 sdU12=.08921 sdU2=1.3483;
* Setting up level-2 equations;
b0i = fasypmp + U0i;
b1i = famount + U1i;
b2i = frate + U2i;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*time1));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL rt ~ normal(PredY , sdE*sdE);
* Defining random effects: normally distributed with means and variances;
RANDOM U0i U1i U2i ~ normal([0,0,0],[sdU0*sdU0, sdU01*sdU01, sdU1*sdU1,
                                sdU02*sdU02, sdU12*sdU12, sdU2*sdU2]) SUBJECT=PersonID;
* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"                               fasypmp;
ESTIMATE "Fixed Amount"                                 famount;
ESTIMATE "Fixed Rate"                                  frate;
ESTIMATE "Level-1 Residual E Variance"                sdE*sdE;
ESTIMATE "Level-2 Random Asymptote U0 Variance"       sdU0*sdU0;
ESTIMATE "Level-2 Asymptote-Amount U01 Covariance"   sdU01*sdU01;
ESTIMATE "Level-2 Random Amount U1 Variance"          sdU1*sdU1;
ESTIMATE "Level-2 Asymptote-Rate U02 Variance"        sdU02*sdU02;
ESTIMATE "Level-2 Amount-Rate U12 Covariance"         sdU12*sdU12;
ESTIMATE "Level-2 Random Rate U2 Variance"            sdU2*sdU2;
* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"                            fasypmp+famount;
ESTIMATE "Session=1 Time=0 Predicted Mean"           fasypmp+(famount*EXP(frate*0));
ESTIMATE "Session=2 Time=1 Predicted Mean"           fasypmp+(famount*EXP(frate*1));
ESTIMATE "Session=3 Time=2 Predicted Mean"           fasypmp+(famount*EXP(frate*2));
ESTIMATE "Session=4 Time=3 Predicted Mean"           fasypmp+(famount*EXP(frate*3));
ESTIMATE "Session=5 Time=4 Predicted Mean"           fasypmp+(famount*EXP(frate*4));
ESTIMATE "Session=6 Time=5 Predicted Mean"           fasypmp+(famount*EXP(frate*5));
RUN; TITLE1; TITLE2;

```

From the log:


ERROR: Quadrature accuracy of 0.000100
could not be achieved with 31 points.
The achieved accuracy was 1.000000.

No convergence, after several tries with different start values and
relaxing the estimation options....

* Simpler estimation to get start values;
PROC NLIN MIXED DATA=work.Example6 METHOD=FIRO;

Since the random rate parameter will not estimate, we will call this “as done as we can be”.

So how did we do? Let’s compare model predictions in terms of means (top) and variances (bottom)?



The absolute fit of the exponential model for the means can be tested by mimicking a saturated means model using the same random asymptotes and amounts (i.e., holding the model for the variance constant):

```
TITLE1 "SAS Test Absolute Fit of Means Model (Using Random Asymptote and Amount Variance Model)";
PROC NLIN MIXED DATA=work.Example6 METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances (in SD metric);
PARMS fasymp=1675 famount=284 frate=-.7 fs1=0 fs2=0 fs3=0
      sdE=184 sdU0=447 sdU01=1 sdU1=10;
* Setting up level-2 equations;
b0i = fasymp + U0i;
b1i = famount + U1i;
b2i = frate;
* Setting up level-1 equation WITHOUT level-1 residual;
* Adding session-specific contrasts;
PredY = (b0i) + (b1i*EXP(b2i*timel)) + (fs1*s1) + (fs2*s2) + (fs3*s3);
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "y" and a variance of "VarE";
MODEL rt ~ normal(PredY, sdE*sdE);
```

```
* Defining random effects: normally distributed with means and variances;
  RANDOM U0i U1i ~ normal([0,0],[sdU0*sdU0,sdU01*sdU01, sdU1*sdU1]) SUBJECT=PersonID;
* Multivariate Wald test for extra session contrasts;
  CONTRAST "Does fixed exponential reproduce saturated means?" fs1*1, fs2*1, fs3*1;
RUN; TITLE1;
```

Fit Statistics	
-2 Log Likelihood	8325.9
AIC (smaller is better)	8345.9
AICC (smaller is better)	8346.3
BIC (smaller is better)	8372.1

Because there are now **6 fixed effects for the 6 means**, this model is equivalent to **saturated means** (even if the fixed slopes are largely uninterpretable). In an SEM context, this model would be specified by letting three of the observed occasions' intercepts be estimated (as discrepancies).

Parameter Estimates							
Parameter	Estimate	Standard	DF	t Value	Pr > t	95% Confidence	
						Limits	Gradient
fasymp	1674.50	46.6780	99	35.87	<.0001	1581.88 1767.12	3.06E-13
famount	743.71	580.66	99	1.28	0.2033	-408.45 1895.87	-342E-15
frate	-0.9188	0.2387	99	-3.85	0.0002	-1.3925 -0.4451	-1.03E-9
fs1	-456.32	576.41	99	-0.79	0.4305	-1600.03 687.40	-14E-13
fs2	-156.07	178.71	99	-0.87	0.3846	-510.66 198.52	-223E-14
fs3	-42.8634	54.9731	99	-0.78	0.4374	-151.94 66.2153	1.31E-12
sdE	151.35	5.3270	99	28.41	<.0001	140.78 161.92	-303E-13
sdU0	438.52	31.9225	99	13.74	<.0001	375.18 501.86	-627E-16
sdU01	90.0473	80.6988	99	1.12	0.2672	-70.0766 250.17	4.48E-13
sdU1	273.57	28.1041	99	9.73	<.0001	217.81 329.34	5.66E-12

The multivariate Wald test using CONTRAST indicates that the 3 extra session contrasts did not improve model fit (which is good news here).

Label	Contrasts		Num DF	Den DF	F Value	Pr > F	Chi-Sq = F*DFnum
Does fixed exponential reproduce saturated means?			3	99	0.29	0.8308	0.87

Given that this is now a saturated means model, we can compare the fit of its exponential variance model (random asymptote + random amount) to an Unstructured R model estimated in ML instead of REML in MIXED.

```
TITLE1 "SAS Ch 6: 0: Saturated Means, Unstructured Variance Model using ML";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=ML;
  CLASS PersonID session;
  MODEL rt = session / SOLUTION CL DDFM=Satterthwaite;
  REPEATED session / TYPE=UN SUBJECT=PersonID;
  ODS OUTPUT InfoCrit=FitMLSatUN;
RUN; TITLE1;
```

Fit Statistics from UN Variance Model (21 parms)	
-2 Log Likelihood	8278.1
AIC (Smaller is Better)	8332.1
AICC (Smaller is Better)	8334.7
BIC (Smaller is Better)	8402.7

$-2\Delta LL(17) = 46.5, p < .001$, so the exponential variance model does not fit (likely because it only has two random effects, whereas all the others we tried in Example6a and Example6b had three).